



# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-272
TITLE	IOPACK - A MESSAGE AND NUMBER I-O UTILITY PACKAGE
AUTHOR	Brian Barton and Kurt Metzger
COMPANY	University of Michigan Ann Arbor, Michigan
DATE	March 5, 1970
SOURCE LANGUAGE	MACRO-8



## IOPACK

DECUS Program Library Write-up

DECUS No. 8-272

### ABSTRACT

IOPACK is one page long and contains: MESSAGE for printing messages; OCTIN and DECIN for inputting unsigned octal or decimal numbers; OCTOUT and DECOUT for outputting unsigned octal or decimal numbers.

### REQUIREMENTS

Family-of-eight computer with teletype.

#### Storage

200<sub>8</sub> locations.

### USAGE

MESSAGE is called using a JMS with the address of a Macro-8 produced text string contained in the AC. This routine uses a % to produce a carriage-return and a # to produce a line-feed. The return is made to the location following the call to MESSAGE with the AC and LINK set to zero. Note: the text does not follow the call to MESSAGE.

OCTIN, DECIN are called using a JMS. The user is to enter either an octal or decimal number from the keyboard. Illegal ASCII character values above 300<sub>8</sub> restart these routines and cause a



back-arrow to be typed. Those values below  $300_8$  terminate the input. The terminating character is not echoed and is stored in location 0 of the page IOPACK is on (MESSAGE's entry point). The return is made to the location following the call with the resultant number (DECIN converts decimal to binary) contained in the AC and with a clear LINK. Input values should not exceed the capacity of one twelve-bit computer word (7777 octal or 4095 decimal) otherwise the results are uncertain.

OCTOUT, DECOUT are called using a JMS. Four digits are typed (DECOUT converts binary to decimal). The return is made to the location following the call with a clear AC and LINK.

Location 24 is assumed to contain the address of teleprinter support routine which is assumed to return with a clear AC. Location 25 is assumed to contain the address of a reader support routine which returns with input characters contained in the AC.

#### RESTRICTIONS

This set of routines is page relocatable.

/UNSIGNED I-O UTILITY PACKAGE

/

/CONTAINS:

/ MESSAGE

/ OCTIN

/ DECIN

/ OCTOUT

/ DECOUT

/

/BRIAN BARTON AND KURT METZGER

/COOLEY ELECTRONICS LABORATORY

/UNIVERSITY OF MICHIGAN

/ANN ARBOR, MICHIGAN 48105

/ MARCH 5, 1970

/

\*1200

/

/MESSAGE ROUTINE

/

/ USE: TAD (MLOC /MLOC=ADDRESS OF TEXT

/ JMS MESSAGE

/

/RETS HERE WITH C(AC)=C(L)=0

/

1200 0000 MESSAGE, 0 /AC=ADDR OF TEXT

1201 3234 DCA OCTIN /USE AS A TEMP

1202 1634 TAD I OCTIN /GET WORD

1203 7012 RTR

1204 7012 RTR

1205 7012 RTR

1206 4213 JMS TYPCH /TYPE LH

1207 1634 TAD I OCTIN

1210 4213 JMS TYPCH /TYPE RH

1211 2234 ISZ OCTIN /ADV PTR

1212 5202 JMP MESSAGE+2

/

1213 0000 TYPCH, 0 /TYPES

1214 7100 CLL /SO EXITS WITH CLEAR LINK

1215 0377 AND (77

1216 7450 SNA /O TERMINATES

1217 5600 JMP I MESSAGE /RETURN

1220 1376 TAD (-40

1221 7510 SPA /WILL BE A 200 CODE

1222 1375 TAD (100 /WILL BE A 300 CODE

1223 1374 TAD (-5 /CHECK FOR CR %

1224 7450 SNA

1225 1373 TAD (-30 /ENDS UP AS 215

1226 1372 TAD (2 /SEE IF LINE FEED #

1227 7450 SNA

1230 1371 TAD (-31 /END UP AS 212

1231 1370 TAD (243 /GET ASCII

1232 4424 JMS I TYPE /TYPE IT

1233 5613 JMP I TYPCH /END

/

/

/UNSIGNED OCTAL INPUT



```

/
/  USE:  JMS OCTIN
/
/                                /RETS HERE WITH C(AC)=VALUE C(L)=0
/
/ THE TERMINATING CHARACTER IS SAVE IN C(MESSAGE)
/   CODES BELOW 300 (OCTAL) TERMINATE
/   CODES ABOVE 277 (OCTAL) RESTART ROUTINE
/   RESTART IS INDICATED BY * BEING TYPED
/

1234 0000 OCTIN, 0 /OCTAL INPUT, 4 DIGITS MAX
1235 7300 CLA CLL
1236 1355 OVERM, TAD OCTTAB /TAD (NOP
1237 3262 DCA DFX /DO OCTAL
1240 1357 TAD OCTTAB+2 /TAD (-10 VAL FOR OCTAL TEST
1241 3304 DCA DECIN /USE AS TEMP
1242 3213 DCA OCTOPT /O OUTPUT VALUE
1243 4425 OCTA, JMS I READ /GET CHARACTER
1244 0367 AND (177
1245 1366 TAD (200 /FORCE PARITY
1246 3200 DCA OCTINP /SAVE
1247 1200 TAD OCTINP
1250 1365 TAD (-260 /SEE IF .GE. 260
1251 7510 SPA /SKIP IF IT MIGHT BE VALID
1252 5272 JMP OCTC /TERMINATE
1253 1304 TAD DECIN /SEE IF IN PROPER RANGE
1254 7700 SMA CLA /SKIP IF OK
1255 5272 JMP OCTC /GREATER THAN INTEGERS
1256 1200 TAD OCTINP
1257 4424 JMS I TYPE
1260 1213 TAD OCTOPT /MULTIPLY BY 8
1261 7106 CLL RTL /MULT BY 4
1262 7000 DFX, NOP /OR TAD OCTOPT FOR DECIMAL
1263 7104 CLL RAL /MULT BY 2
1264 3213 DCA OCTOPT
1265 1200 TAD OCTINP
1266 1365 TAD (-260 /GET VALID VALUE
1267 1213 TAD OCTOPT /UPDATE VALUE
1270 3213 DCA OCTOPT
1271 5243 JMP OCTA /MORE ALLOWED
1272 7300 OCTC, CLA CLL
1273 1200 TAD OCTINP
1274 1364 TAD (-300 /BELOW 300 TERMINATES
1275 7710 SPA CLA /ABOVE IS * I.E. KILL NUMBER
1276 5302 JMP OVERL
1277 1363 TAD (337 /~
1300 4424 JMS I TYPE
1301 5242 JMP OCTA-1 /RESTART
1302 1213 OVERL, TAD OCTOPT /USED AS A OVERLAY CONSTANT
1303 5634 JMP I OCTIN /RET W AC=VALUE L=0
/
/
/ UNSIGNED DECIMAL INPUT ROUTINE
/
/  USE:  JMS DECIN
/
/                                /RETS HERE WITH C(AC)=VALUE C(L)=0
/

```

/SEE NOTES FOR OCTIN ON TERMINATORS  
/

```

1304 0000 DECIN, 0
1305 7300 CLA CLL
1306 1304 TAD DECIN /SET UP OCTIN FOR DECIMAL
1307 3234 DCA OCTIN /FIX RET POINT
1310 1302 TAD OVERL /TAD (TAD OCTOPT
1311 3262 DCA DFX /SET UP OVERLAY
1312 7344 STA CLL RAL /AC=-2
1313 5240 JMP OCTA-3 /ENTER OCTIN
/
/
/UNSIGNED OCTAL OUTPUT ROUTINE
/
/ USE: TAD VALUE
/ JMS OCTOUT
/ /RETS HERE WITH C(AC)=C(L)=0
/

```

```

1314 0000 OCTOUT, 0 /OCTAL OUTPUT..UNSIGNED
1315 3304 DCA DECIN /SAVE VALUE
1316 1236 TAD OVERM /TAD (TAD OCTTAB
1317 3325 DCA OCTPTR /SET UP DIVIDE TABLE LOC
1320 1362 TAD (-4
1321 3343 DCA DECOUT /USE AS A COUNTER
1322 3200 OCTLA, DCA MESSAGE /USE AS A DIVIDEND
1323 1304 TAD DECIN /GET VALUE
1324 7100 CLL /USE LINK TO FIND OVERFLOW
1325 1355 OCTPTR, TAD OCTTAB /OR TAD DECTAB
1326 7420 SNL /L=0 MEANS LOOP DONE
1327 5333 JMP .+4
1330 2200 ISZ MESSAGE /ADD 1 TO DIVIDEND
1331 3304 DCA DECIN /SAVE REMAINDER
1332 5323 JMP OCTLA+1 /SUBTRACT AGAIN
1333 7300 CLA CLL
1334 1200 TAD MESSAGE /HAVE THE DESIRED DIGIT
1335 1361 TAD (260 /MAKE ASCII
1336 4424 JMS I TYPE
1337 2325 ISZ OCTPTR /ADV PTR IN TABLE
1340 2343 ISZ DECOUT /4 DIGITS?
1341 5322 JMP OCTLA /NO
1342 5714 JMP I OCTOUT /YES, EXIT WITH C(AC)=C(L)=0
/
/

```

/UNSIGNED DECIMAL OUTPUT ROUTINE  
/  
/ USE: TAD VALUE  
/ JMS DECOUT  
/ /RET HERE WITH C(AC)=C(L)=0  
/

```

1343 0000 DECOUT, 0 /DECIMAL OUTPUT..UNSIGNED
1344 3304 DCA DECIN /OUTPUT THIS VALUE
1345 1343 TAD DECOUT /USE OCTOUT ROUTINE
1346 3314 DCA OCTOUT
1347 1362 TAD (-4 /USE DECIMAL TABLE
1350 5316 JMP OCTOUT+2
/

```



1351	6030	6030 /-1000 DECIMAL
1352	7634	7634 /-100 DECIMAL
1353	7766	7766 /-10 DECIMAL
1354	7777	7777 /-1 DECIMAL
1355	7000	OCTTAB, 7000 /-1000 OCTAL
1356	7700	7700 /-100 OCTAL
1357	7770	7770 /-10 OCTAL
1360	7777	7777 /-1 OCTAL

/

OCTINP=MESAGE

OCTOPT=TYPCH

TYPE=24 /ADDR OF TYPE ROUTINE..RETS WITH AC=0

READ=25 /ADDR OF READ ROUTINE..RETS WITH ASCII IN AC

/

1361	0260	PAGE
1362	7774	
1363	0337	
1364	7500	
1365	7520	
1366	0200	
1367	0177	
1370	0243	
1371	7747	
1372	0002	
1373	7750	
1374	7773	
1375	0100	
1376	7740	
1377	0077	